Analysis of Ecological Momentary Assessment Data Using Multilevel Modeling

Michael Russell
The Methodology Center
For C-StARR Writing Group, Duke & UNC
November 17, 2016

Repeated Assessments Give Us Multiple Levels of Information



Example: Craving Among Smokers

- Craving as a primary driver of smoking behavior among smokers
- Craving may differ *within people* over time
  - *dynamic,* can be "triggered" by momentary factors: e.g., time of day, negative affect
- Craving may also differ *between people,* on average
  - Some people crave nicotine more than others
  - Driven by person-level characteristics, e.g. dependence level

Example Study Design

- 1000 Smokers, no plans to quit

- Assessed 5 times daily
  - Evenly spaced from Morning to Evening
  - *Measures focus on craving, affect (positive and negative)*

- Baseline Measures obtained before EMA
  - *Dependence level, childhood adversity*


How much have you been bothered by the desire to smoke a cigarette **since the last survey**?
9
Not at all
Very much
Previous
Next

PennState

Craving varies both between smokers, and within smokers over time



Roadmap

1. Intro to MLM and Partitioning Between versus Within Variance in Craving
2. Craving by Time of Day
3. Craving Time Course by Baseline Dependence
4. Modeling Within-Person Processes: Negative Affect and Craving
5. Within-Person Process by Between-Person Characteristics

Intro to Multilevel Modeling

PennState

Multilevel Models for EMA Data

- Multilevel models (MLM) are frequently used in EMA analyses

- Features:
  - Allow prediction of outcome (craving) at multiple levels
  - Adjust for non-independence of observations due to repeated measurement of outcome over time

Multilevel models also known as…

- Mixed models
- Random effects models
- General linear mixed models
- Hierarchical linear models (HLM)
- Growth curve models (special case of MLM)

What are they, in a nutshell?

- The term "mixed" in mixed models (and SAS PROC MIXED) refers to a generalization of standard linear regression
  - Allows for a "mix" of both fixed and random effects

- *Fixed Effects* refer to the regression coefficients (intercepts and slopes)

- *Random Effects* refer to variance around these coefficients (intercept variance, slope variance, residual variance)

Why use MLM? EMA data are "nested"

- If you collect repeated measurements on a set of individuals, time is *nested* within individual
- Nesting causes dependence: Two assessments from same individual more similar than two assessments from different individuals
- Multilevel models account for this:
  - Repeated assessments at Level 1 (within-person)
  - Individuals at Level 2 (between-person)

What are the levels?

- Level 1 is the **Assessment-Level Model**
  - We assume Y_{ij} (outcome for time i, person j) is a function of assessment-specific characteristics plus random error
  - *"WHEN" questions*
  - *Is craving higher <u>when</u> negative affect is high?*
- Level 2 is the **Individual-Level Model**
  - Parameters in Level 1 (effects of assessment-specific characteristics on outcome) vary across individuals; parameters modeled as function of individual characteristics
  - *"WHO" questions*
  - *Is average craving higher for smokers <u>who</u> show higher dependence?*

PennState

Partitioning Between versus Within Variance in Craving

Partitioning Variance with MLM

- Let's start with a simple MLM
- We are interested in *how much* of the variation in craving is *between* versus *within*
- We specify an "empty" MLM – a model with no predictors – to figure this out

Empty MLM for Craving

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + u_{0j}$

Level 1 is the Assessment level, modeling moment-to-moment relationships within a person (why craving differs from moment to moment within the same person)

Level 2 is the Person level, modeling relationships between people (why average craving levels differ between people)

Mixed Equation…What Proc Mixed Uses

Multilevel Equation:

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + e_{ij}$

LEVEL 2: $\boxed{\beta_{0j} = \gamma_{00} + u_{0j}}$

Mixed Equation:

$\mathbf{CRAV}_{ij} = \boxed{\gamma_{00} + u_{0j}} + e_{ij}$

In the mixed equation, we substitute β_{0j} for the pieces that make it up:

- γ_{00} , the fixed intercept
- u_{0j}, the random intercept

PennState

PROC MIXED for Empty Model

```
title "Craving Empty Model";
proc mixed data=ILDDataset method=ml covtest;
class id;
model craving = /solution cl ddfm=bw;
random intercept /subject=id type=un g;
repeated /sub=id type=vc;
run; title;
```

Model: specifies fixed effects (intercepts and slopes)
Random: specifies random effects (variances of intercepts and slopes)
Repeated: specifies within-person residual variance

PROC MIXED for Empty Model

```
TITLE "Craving Empty Model";
PROC MIXED data=ILDDataset METHOD=ml COVTEST;
CLASS id;
MODEL craving = /SOLUTION CL DDFM=bw;
RANDOM intercept /SUBJECT=id TYPE=un G;
REPEATED /SUB=id TYPE=vc;
RUN; TITLE;
```

What are we estimating?

$\mathbf{CRAV}_{ij} = \gamma_{00} + u_{0j} + e_{ij}$

Model: Intercept fixed effect for craving (γ_{00})
Random: Craving intercept variance (Variance of u_{0j})
Repeated: Level 1 residual variance (Variance of e_{ij})

Selected Model Output

Fixed Effects

- The Grand Mean of Craving

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Intercept	5.0076	0.08157	999	61.39	<.0001	0.05	4.8476	5.1677

Random Effects

- Between-person variance in craving
- Within-person variance in craving

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
Intercept	id	6.5218	0.2975	21.92	<.0001
Residual	id	3.9396	0.03248	121.31	<.0001

Interpretations

What is the mean of craving across all people and observations?

Solution for Fixed Effects								
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Intercept	5.0076	0.08157	999	61.39	<.0001	0.05	4.8476	5.1677

Do some people have higher mean craving levels than others?

Does craving vary from moment to moment within a person?

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
Intercept	id	6.5218	0.2975	21.92	<.0001
Residual	id	3.9396	0.03248	121.31	<.0001

Partitioning Variance via the Intraclass Correlation (ICC)

- The ICC is the proportion of variance in the outcome that is between (vs within) people
- Can be thought of as a "clustering coefficient", average correlation between repeated observations
- The ICC is

The Amount of Between-Person Variance

divided by

The Total Variance (between + within)

Craving ICC

Between-person variance

Within-person variance

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
Intercept	id	6.5218	0.2975	21.92	<.0001
Residual	id	3.9396	0.03248	121.31	<.0001

ICC = var(Intercept) / (var(Intercept) + var(Residual))
= 6.52 / (6.52 + 3.94)
= 0.62

62% of the variance in craving is <u>between people</u>

1-ICC gives the variance *<u>within people</u>* *(1-0.62 = 0.38)*

The remaining 38% of the variance in craving is within people over time.

PennState

Craving by Time of Day

PennState

Modeling the Time Trends in Craving

- Over a third of the variance in craving is at the within-person level
- What might cause within-person fluctuation in craving?
- We suspect craving in smokers may vary by time
  - Time of day (morning, afternoon, evening)

Estimating the relationship

- We are interested in modeling how craving varies from the start to the end of a day
- We create a variable that counts the hours since the start of the day, from 8 AM to 8 PM
- We estimate the relationship using MLM

MLM for Craving by Time-of-Day

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{TODc}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + u_{0j}$

$\beta_{1j} = \gamma_{10}$

Time of day (TOD) is centered ("c") at 8 AM.

β_{1j} is the linear slope describing the relationship between craving and time of day

β_{0j} is the intercept, the predicted level of craving at 8 AM

PennState

Mixed Model for Craving by Time-of-Day

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{10}(\mathbf{TODc}_{ij}) \quad \leftarrow \textit{Fixed effects}$$
$$+ u_{0j} + e_{ij} \quad \leftarrow \textit{Random effects}$$

- γ_{00} : fixed intercept
- γ_{10} : fixed time-of-day slope
- u_{0j} : random intercept
- e_{ij} : residual

PennState

SAS Proc Mixed Syntax

```
title "Time-of-Day Association with Craving";
proc mixed data=ILDDataset method=ml covtest;
 class id;
 model Craving= hourscontc /solution ddfm=bw;
 random intercept /subject=id type=un g gcorr;
 repeated /sub=id type=vc;
run; title;
```

Selected Output

Fixed Effects

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.6810	0.08455	999	67.19	<.0001
hourscontc	-0.1012	0.003329	29E3	-30.38	<.0001

Random Effects

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	id	6.5312	0.2978	21.93	<.0001
Residual	id	3.8197	0.03149	121.31	<.0001

Autocorrelation in Residuals?

- Models so far make the assumption that residuals are independent (correlation = 0) within an individual

	t_1	t_2	t_3	t_4
t_1	1	0	0	0
t_2	0	1	0	0
t_3	0	0	1	0
t_4	0	0	0	1

- May not be reasonable.
  - Measures taken closer in time are likely to be more similar to those farther apart

PennState

Autoregressive Lag 1

- An autoregressive lag 1 [AR(1)] structure is often used in daily diaries
- Assumes decomposing correlation structure *by observation number*, and assumes equal spacing of observations over time

	t_1	t_2	t_3	t_4
t_1	1	ρ	ρ^2	ρ^3
t_2	ρ	1	ρ	ρ^2
t_3	ρ^2	ρ	1	ρ
t_4	ρ^3	ρ^2	ρ	1

Spatial Power [SP(POW)]

- Equal spacing of observations is often not a good assumption for EMA designs

- The Spatial Power structure is an adapted version of AR(1)
  - Assumes decomposing correlation, but weights lagged correlations by the *time distance* between observations (d_{ij})

	t_1	t_2	t_3	t_4
t_1	1	ρ^{d12}	ρ^{d13}	ρ^{d14}
t_2	ρ^{d21}	1	ρ^{d23}	ρ^{d24}
t_3	ρ^{d31}	ρ^{d32}	1	ρ^{d34}
t_4	ρ^{d41}	ρ^{d42}	ρ^{d43}	1

PennState

Adjusting our Model for SP(POW) error autocorrelation

```
title "Time-of-Day Association with Craving, SP(POW) Residual
Autocorrelation";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= hourscontc /solution ddfm=bw;
random intercept /subject=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

Significant autocorrelation, little difference in parameters

Fixed Effects, without SP(POW)

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.6810	0.08455	999	67.19	<.0001
hourscontc	-0.1012	0.003329	29E3	-30.38	<.0001

Fixed Effects, with SP(POW)

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.6981	0.08454	999	67.40	<.0001
hourscontc	-0.1017	0.003403	29E3	-29.89	<.0001

Random Effects, without SP(POW)

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	id	6.5312	0.2978	21.93	<.0001
Residual	id	3.8197	0.03149	121.31	<.0001

Random Effects, with SP(POW)

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	6.4970	0.2967	21.90	<.0001
SP(POW)	id	0.9731	0.001385	702.59	<.0001
Residual		3.8465	0.03217	119.58	<.0001

PennState

Random Effect?

- We find that craving appears to decrease throughout the day
  - But does it work the same way for everyone?
- Perhaps we think that the link between time of day and craving differs randomly from person to person
- Build in a random slope for time of day



Craving by time of day
10
11
12
13
14
15
16
17
Craving
Time of Day (in hours)

MLM for Craving by Time of Day, with Random Slope

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{TODc}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + u_{0j}$

$\beta_{1j} = \gamma_{10} + u_{1j}$

PennState

Mixed Model for Craving by Time of Day, with Random Slope

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{10}(\mathbf{TODc}_{ij}) + u_{0j} + u_{1j}(\mathbf{TODc}_{ij}) + e_{ij}$$

γ_{00} : fixed intercept

γ_{10} : fixed time-of-day slope

u_{0j} : random intercept

u_{1j} : random time-of-day slope

e_{ij} : residual

SAS Proc Mixed Syntax

```
title "Time-of-Day Association with Craving, random slope";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= hourscontc /solution ddfm=bw;
random intercept hourscontc /subject=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

PennState

Selected Output

Fixed Effects

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.6974	0.08559	999	66.57	<.0001
hourscontc	-0.1017	0.003519	29E3	-28.90	<.0001

Random Effects

Estimated G Matrix				
Row	Effect	id	Col1	Col2
1	Intercept	1	6.6766	-0.01617
2	hourscontc	1	-0.01617	0.000826

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	6.6766	0.3279	20.36	<.0001
UN(2,1)	id	-0.01617	0.009983	-1.62	0.1052
UN(2,2)	id	0.000826	0.000556	1.49	0.0686
SP(POW)	id	0.9728	0.001421	684.52	<.0001
Residual		3.8363	0.03271	117.29	<.0001

Extending the Model to a Quadratic

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{10}(\mathbf{TODc}_{ij}) + \gamma_{20}(\mathbf{TODc}_{ij}{}^{2}) + u_{0j} + u_{1j}(\mathbf{TODc}_{ij}) + u_{2j}(\mathbf{TODc}_{ij}{}^{2}) + e_{ij}$$

γ_{00} : fixed intercept
γ_{10} : fixed time-of-day linear slope
γ_{20} : fixed time-of-day quadratic
u_{0j} : random intercept
u_{1j} : random time-of-day linear slope
u_{2j} : random time-of-day quadratic
e_{ij} : residual

Extending the Model to a Quadratic

```
title "Time-of-Day Association with Craving, Random Slope & Quadratic";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= hourscontc hourscontcsq /s ddfm=bw;
random intercept hourscontc hourscontcsq /sub=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

Output, Quadratic Model

Fixed Effects

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	7.7809	0.07841	999	99.24	<.0001
hourscontc	-0.9508	0.01764	29E3	-53.91	<.0001
hourscontcsq	0.06382	0.001292	29E3	49.39	<.0001

Random Effects

Estimated G Matrix					
Row	Effect	id	Col1	Col2	Col3
1	Intercept	1	4.4211	0.3250	-0.02671
2	hourscontc	1	0.3250	0.1084	-0.00786
3	hourscontcsq	1	-0.02671	-0.00786	0.000583

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	4.4211	0.2726	16.22	<.0001
UN(2,1)	id	0.3250	0.04359	7.46	<.0001
UN(2,2)	id	0.1084	0.01413	7.67	<.0001
UN(3,1)	id	-0.02671	0.003181	-8.40	<.0001
UN(3,2)	id	-0.00786	0.001027	-7.66	<.0001
UN(3,3)	id	0.000583	0.000076	7.66	<.0001
SP(POW)	id	0.9414	0.01296	72.63	<.0001
Residual		3.3069	0.02839	116.49	<.0001

PennState



$$CRAV_{ij} = 7.78 - 0.95(TODc) + 0.06(TODc^2)$$

Conclusions from Time-Of-Day Model

- Craving starts high in the morning, reaches a low in the early afternoon, slight increase in the evening

- Random effects show significant heterogeneity in the daily time course of craving

Craving Time Course by Baseline Dependence

Between-Person Effects

- Smokers vary in their level of nicotine dependence
- These differences in dependence may influence their average levels of craving
- As well as their craving dynamics

Does average craving differ by Baseline Dependence?

- We hypothesize that smokers higher in dependence will experience more intense craving on average.
- Dependence
  - Measured via the item "*How soon after you wake up do you smoke your first cigarette?*"
    - Within 5 minutes, 5-30 mins, 31-60 mins, Over 60 mins
    - Dichotomized into high (within 5 minutes, 9%) versus low-to-moderate dependence (6-60+ mins, 91%)

MLM for Dependence Effect

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + \gamma_{01}(\mathrm{DEP}_j) + u_{0j}$

γ_{00} is the intercept, the predicted level of craving when Dependence = 0 (low-to-moderate dependence)

γ_{01} is the Dependence association, or the increase in craving when Dependence = 1 (high dependence)

PennState

Mixed Model for Dependence Effect

Multilevel Equation:

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + e_{ij}$

LEVEL 2: $\boxed{\beta_{0j} = \gamma_{00} + \gamma_{01}(\mathrm{DEP}_j) + u_{0j}}$

Mixed Equation:

$\mathbf{CRAV}_{ij} = \boxed{\gamma_{00} + \gamma_{01}(\mathrm{DEP}_j) + u_{0j}} + e_{ij}$

γ_{00} : fixed intercept

γ_{01} : Dependence effect on intercept

u_{0j} : random intercept

e_{ij} : residual

SAS Syntax for Between Model

```
title "Dependence association with craving";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= ftnd0 /s ddfm=bw;
random intercept /subject=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

Selected Model Results

Fixed Effects:

Those with high dependence show higher mean craving than those with lower dependence

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	4.5990	0.07476	998	61.52	<.0001
ftnd0	4.1984	0.2353	998	17.84	<.0001

Random Effects

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	4.8820	0.2247	21.73	<.0001
SP(POW)	id	0.9748	0.001176	829.14	<.0001
Residual		3.9711	0.03327	119.36	<.0001

PennState

Dependence as a Predictor of Craving Dynamics

- We might also hypothesize that dependence level changes the dynamics of craving
- Perhaps craving for those with high dependence is less dependent on time of day compared to those with low-to-moderate dependence
- Add dependence to our time-of-day MLM

MLM for Time-of-Day by Dependence

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{TODc}_{ij}) + \beta_{2j}(\mathbf{TODc}_{ij}{}^{2}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_{j}) + u_{0j}$

$\beta_{1j} = \gamma_{10} + \gamma_{11}(\mathbf{DEP}_{j}) + u_{1j}$

$\beta_{2j} = \gamma_{20} + \gamma_{21}(\mathbf{DEP}_{j}) + u_{2j}$

Mixed Model for Time-of-Day by Dependence

$$\begin{aligned}\mathbf{CRAV}_{ij} = {} & \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j) \\ & + \gamma_{10}(\mathbf{TODc}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \text{ x } \mathbf{TODc}_{ij}) \\ & + \gamma_{20}(\mathbf{TODc}_{ij}{}^2) + \gamma_{21}(\mathbf{DEP}_j \text{ x } \mathbf{TODc}_{ij}{}^2) \\ & + u_{0j} + u_{1j}(\mathbf{TODc}_{ij}) + u_{2j}(\mathbf{TODc}_{ij}{}^2) + e_{ij}\end{aligned}$$

$\gamma_{11}(\mathbf{DEP}_j \text{ x } \mathbf{TODc}_{ij})$ and $\gamma_{21}(\mathbf{DEP}_j \text{ x } \mathbf{TODc}_{ij}{}^2)$ are *cross-level interactions*, where a level 2 moderator (between-person) is interacted with a level 1 predictor (within person)

SAS Syntax

```
title "Craving by time of day Quadratic x Dependence";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= ftnd0
                    hourscontc ftnd0*hourscontc
                    hourscontcsq ftnd0*hourscontcsq /s ddfm=bw;
random intercept hourscontc hourscontcsq
          /subject=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

Output

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	7.5255	0.07875	998	95.56	<.0001
hourscontc	-1.0176	0.01735	29E3	-58.65	<.0001
hourscontcsq	0.06876	0.001270	29E3	54.15	<.0001
ftnd0	2.5333	0.2478	998	10.22	<.0001
hourscontc*ftnd0	0.6616	0.05465	29E3	12.11	<.0001
hourscontcsq*ftnd0	-0.04891	0.004003	29E3	-12.22	<.0001

Estimated G Matrix					
Row	Effect	id	Col1	Col2	Col3
1	Intercept	1	3.8529	0.1715	-0.01539
2	hourscontc	1	0.1715	0.06841	-0.00491
3	hourscontcsq	1	-0.01539	-0.00491	0.000365

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	3.8529	0.2475	15.57	<.0001
UN(2,1)	id	0.1715	0.03985	4.30	<.0001
UN(2,2)	id	0.06841	0.01224	5.59	<.0001
UN(3,1)	id	-0.01539	0.002876	-5.35	<.0001
UN(3,2)	id	-0.00491	0.000887	-5.54	<.0001
UN(3,3)	id	0.000365	0.000066	5.55	<.0001
SP(POW)	id	0.9411	0.01320	71.31	<.0001
Residual		3.3065	0.02838	116.52	<.0001

Creating trajectories by group

- All parts of the curve differ by dependence group (intercept, slope, quadratic)

- We can generate a curve for each group, and discover intercept, slope, and quadratic values at low and high dependence

PennState

Use Fixed Effects Equation to Generate Predicted Values

$$\begin{aligned}\textbf{Pred\_CRAV}_{ij} = {} & \gamma_{00} + \gamma_{01}(\textbf{DEP}_j) \\ & + \gamma_{10}(\textbf{TODc}_{ij}) + \gamma_{11}(\textbf{DEP}_j \text{ x } \textbf{TODc}_{ij}) \\ & + \gamma_{20}(\textbf{TODc}_{ij}{}^2) + \gamma_{21}(\textbf{DEP}_j \text{ x } \textbf{TODc}_{ij}{}^2)\end{aligned}$$

PennState

Low-to-Mod Dependence (DEP=0), 9 AM (TODc=1)

$$\textbf{Pred\_CRAV}_{ij} = 7.53 + 2.53(\mathbf{0}) + (-1.02*(\mathbf{1})) + 0.66*(\mathbf{0} \times \mathbf{1}) + 0.07*(\mathbf{1}) + (-0.05*(\mathbf{0} \times \mathbf{1}))$$

$$\textbf{CRAV}_{ij} = (7.53) + (-1.02 *(1)) + (0.07*(1))$$

$$\textbf{CRAV}_{ij} = 6.58$$

High Dependence (DEP=1), 9 AM (TODc=1)

$$\textbf{Pred\_CRAV}_{ij} = 7.53 + 2.53(\mathbf{1}) + (-1.02*(\mathbf{1})) + 0.66*(1 \times \mathbf{1}) + 0.07*(\mathbf{1}) + (-0.05*(\mathbf{1} \times \mathbf{1}))$$

$$\textbf{CRAV}_{ij} = (10.06) + (-0.36) + (0.02)$$

$$\textbf{CRAV}_{ij} = 9.72$$

Getting estimates from the model via ESTIMATE statements

- Estimate statements allow us to generate point estimates for graphing

- And group-specific intercepts, slopes, and quadratics

Programming ESTIMATE statements

```
title "Craving by time of day Quadratic x FTND, points for graphing";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= ftnd0
                        hourscontc ftnd0*hourscontc
                  hourscontcsq ftnd0*hourscontcsq /s ddfm=bw;
random intercept hourscontc hourscontcsq /subject=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
estimate "9 AM, Low Dep" intercept 1 hourscontc 1 hourscontcsq 1
                        ftnd0 0 ftnd0*hourscontc 0
            ftnd0*hourscontcsq 0 /cl;
estimate "9 AM, High Dep" intercept 1 hourscontc 1 hourscontcsq 1
                        ftnd0 1 ftnd0*hourscontc 1
            ftnd0*hourscontcsq 1 /cl;
ETC…..;
```

$$
\begin{aligned}
\mathbf{Pred\_CRAV}_{ij} = \gamma_{00} &+ \gamma_{01}(\mathbf{DEP}_j) \\
&+ \gamma_{10}(\mathbf{TODc}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \text{ x } \mathbf{TODc}_{ij}) \\
&+ \gamma_{20}(\mathbf{TODc}_{ij}^{2}) + \gamma_{21}(\mathbf{DEP}_j \text{ x } \mathbf{TODc}_{ij}^{2})
\end{aligned}
$$

```
estimate "9 AM, Low Dep" intercept 1 ftnd0 0
                         hourscontc 1 ftnd0*hourscontc 0
                         hourscontcsq 1 ftnd0*hourscontcsq 0 /cl;
```

$$
\begin{aligned}
\mathbf{Pred\_CRAV}_{ij} = \gamma_{00} &+ \gamma_{01}(\mathbf{0}) \\
&+ \gamma_{10}(\mathbf{1}) + \gamma_{11}(0 \text{ x } \mathbf{1}) \\
&+ \gamma_{20}(\mathbf{1}) + \gamma_{21}(\mathbf{0} \text{ x } \mathbf{1})
\end{aligned}
$$

$$\mathbf{Pred\_CRAV}_{ij} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j)$$
$$+ \gamma_{10}(\mathbf{TODc}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \text{ x } \mathbf{TODc}_{ij})$$
$$+ \gamma_{20}(\mathbf{TODc}_{ij}{}^2) + \gamma_{21}(\mathbf{DEP}_j \text{ x } \mathbf{TODc}_{ij}{}^2)$$

```
estimate "9 AM, High Dep" intercept 1 ftnd0 1
                          hourscontc 1 ftnd0*hourscontc 1
                          hourscontcsq 1 ftnd0*hourscontcsq 1 /cl;
```

$$\mathbf{Pred\_CRAV}_{ij} = \gamma_{00} + \gamma_{01}(\mathbf{1})$$
$$+ \gamma_{10}(\mathbf{1}) + \gamma_{11}(1 \text{ x } \mathbf{1})$$
$$+ \gamma_{20}(\mathbf{1}) + \gamma_{21}(\mathbf{1} \text{ x } \mathbf{1})$$

Simple Effects

- We can also use the equation to estimate simple effects
  - (intercepts, slopes, quadratics by dependence group)

$$\mathbf{Pred\_CRAV}_{ij} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j) + \gamma_{10}(\mathbf{TODc}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \times \mathbf{TODc}_{ij}) + \gamma_{20}(\mathbf{TODc}_{ij}^{2}) + \gamma_{21}(\mathbf{DEP}_j \times \mathbf{TODc}_{ij}^{2})$$

Factoring out TODc and $TODc^2$, we get:

Simple Intercept: $\gamma_{00} + \gamma_{01}(\mathbf{DEP}_j)$

Simple Slope: $\gamma_{10} + \gamma_{11}(\mathbf{DEP}_j)$

Simple Quadratic: $\gamma_{20} + \gamma_{21}(\mathbf{DEP}_j)$

Programming Estimate Statements for Simple Effects

Simple Intercept: $\gamma_{00} + \gamma_{01}(\mathbf{DEP}_j)$

Simple Slope: $\gamma_{10} + \gamma_{11}(\mathbf{DEP}_j)$

Simple Quadratic: $\gamma_{20} + \gamma_{21}(\mathbf{DEP}_j)$

```
*simple equations, Low Dep;
estimate "intercept for Low Dep" intercept 1 ftnd0 0;
estimate "slope for Low Dep" hourscontc 1 ftnd0*hourscontc 0;
estimate "quadratic for Low Dep" hourscontcsq 1 ftnd0*hourscontcsq 0;
*simple equations, High Dep;
estimate "intercept for High Dep" intercept 1 ftnd0 1;
estimate "slope for High Dep" hourscontc 1 ftnd0*hourscontc 1;
estimate "quadratic High Low Dep" hourscontcsq 1 ftnd0*hourscontcsq 1;
```

PennState



<u>High Dep</u>:

$$CRAV_{ij}=10.06 - 0.36(TODc) + 0.02(TODc^2)$$

<u>Low Dep</u>:

$$CRAV_{ij}=7.53 - 1.02(TODc) + 0.07(TODc^2)$$

Modeling Within-Person Processes: Negative Affect and Craving

Modeling Within-Person Process

- So far we've seen how craving changes with time itself
- But we could also see how craving varies in relation to *other dynamic, time-varying factors*
- For example, we might suspect that nicotine craving increases with increased negative affect (NA; sadness, anger, anxiety)


Right now I feel:
Sad
4
Not at all
Extremely
Previous
Next


Right now I feel:
Nervous
6
Not at all
Extremely
Previous
Next


Right now I feel:
Tense
5
Not at all
Extremely
Previous
Next

NA and Craving

- NA varies both between people (some have higher average NA than others) and within people (NA varies from moment to moment)
- Thus, NA and craving could be related at *both* levels of analysis
  - Between-person: Smokers with higher mean NA experience more craving on average
  - Within-person: Comparing each smoker to him- or herself, craving increases when NA is high and decreases when NA is low

Naïve Approach

- If we just estimate a model with NA predicting craving, the association we get is difficult to interpret
- Blends between-person and within-person effects of NA
- Helpful to disaggregate between and within portions of NA variable before modeling

Disaggregating between and within

- First, we estimate the mean of NA for each person (this gives us the BETWEEN variable)
- Second, we subtract the person mean from the raw NA scores for each moment/observation (person-mean centering, gives us the WITHIN variable)
- Third, we use one (or both) in models to discover associations at each level

Disaggregating between and within via person-mean centering

$$NA_{ij} = \overline{NA}_j + \breve{NA}_{ij}$$

Negative affect at time *i* for person *j* (NA_{ij}) can be split into 2 parts:

1. The *between* part, the mean for person *j* ($\overline{NA}_j$)
2. The *within* part, the difference between person *j* mean and value at time *i* ($\breve{NA}_{ij} = NA_{ij} - \overline{NA}_j$)

PennState

What each part predicts

- Person-mean NA ($\overline{NA}_j$) varies *only at the between-person level*, can only predict between-person differences in mean craving.

- Person-mean centered NA ($\breve{NA}_{ij}$) varies *only at the within-person level*, predicts only within-person fluctuations in craving

  - This is because by subtracting person means, each person's mean is set to the same value: 0. Thus, it contains <u>no</u> between-person variance.

Calculating person-mean NA ($\overline{NA}_j$) in SAS

```
*MAKE PERSON-MEAN NEGATIVE AFFECT;
proc means data=ilddataset nway noprint;
class id;
var NegAffectC;
output out=means2 mean=mNegAffectC;
run;

*MERGE PERSON-MEAN AND RAW DATA;
data ilddataset;
merge ilddataset means2;
by id;
drop _type_ _freq_;
label mNegAffectC="negative affect, person-mean for between effect";
run;
```

Create person-mean centered NA in SAS ($\widetilde{NA}_{ij} = NA_{ij} - \overline{NA}_{j}$)

```
*CREATE PERSON-MEAN CENTERED NA FOR WITHIN EFFECT;
data ilddataset;
 set ilddataset means2;
 dNegAffectC = NegAffectC - mNegAffectC;
 label dNegAffectC="negative affect, person-mean centered for within effect";
run;
```

Means and Correlations

Simple Statistics							
Variable	**N**	**Mean**	**Std Dev**	**Sum**	**Minimum**	**Maximum**	**Label**
NegAffectC	30432	0.0007341	0.64825	22.34032	-0.52746	5.47254	Negative Affect, Grand Mean Centered
mNegAffectC	35000	0.00234	0.37188	82.05728	-0.52746	1.43087	negative affect, person-mean for between effect
dNegAffectC	30432	0	0.53228	0	-1.73333	5.41935	negative affect, person-mean centered for within effect

Pearson Correlation Coefficients
Prob > |r| under H0: Rho=0
Number of Observations

	NegAffectC	**mNegAffectC**	**dNegAffectC**
NegAffectC Negative Affect, Grand Mean Centered	1.00000 30432	0.57078 <.0001 30432	0.82110 <.0001 30432
mNegAffectC negative affect, person-mean for between effect	0.57078 <.0001 30432	1.00000 35000	0.00000 1.0000 30432
dNegAffectC negative affect, person-mean centered for within effect	0.82110 <.0001 30432	0.00000 1.0000 30432	1.00000 30432

Within-Person MLM for Negative Affect and Craving

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{dNA}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + u_{0j}$

$\beta_{1j} = \gamma_{10}$

dNA_{ij} is negative affect, person-mean centered. Association gives within-person effect, contains only within-person variation.

Given centering, intercept is predicted craving at grand mean negative affect

PennState

Mixed Model for Negative Affect and Craving

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{10}(\mathbf{dNA}_{ij}) + u_{0j} + e_{ij}$$

γ_{00} : fixed intercept
γ_{10} : within-person negative affect slope
u_{0j} : random intercept
e_{ij} : residual

SAS Syntax

```
title "within association between NA and craving";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= dNegAffect /s ddfm=bw;
random intercept /subject=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

Selected Output

Fixed Effects:

Moments with higher negative affect are associated with higher craving than moments with lower negative affect

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.0173	0.08144	999	61.60	<.0001
dNegAffectC	1.2492	0.02017	29E3	61.92	<.0001

Random Effects

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	6.5104	0.2966	21.95	<.0001
SP(POW)	id	0.9699	0.001781	544.54	<.0001
Residual		3.4984	0.02916	119.96	<.0001

Association differs from person to person?

- Within-person NA→craving association may be stronger for some versus others
  - For some, craving may be more strongly driven by bad mood
  - For others, less so
- We can account for differences via random slopes

Within-Person MLM for Negative Affect and Craving, with Random Slope

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{dNA}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + u_{0j}$

$\beta_{1j} = \gamma_{10} + u_{1j}$

Here we add a heterogeneity term (u_{1j}) to suggest that the dNA slope differs from person to person.

Mixed Model for Negative Affect and Craving , with Random Slope

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{10}(\mathbf{dNA}_{ij}) + u_{0j} + u_{1j}(\mathbf{dNA}_{ij}) + e_{ij}$$

γ_{00} : fixed intercept
γ_{10} : within-person negative affect slope
u_{0j} : random intercept
u_{1j} : random within-person NA slope
e_{ij} : residual

PennState

Setting this up in SAS

```
title "within association between NA and craving, with random slope";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= dNegAffectC /s ddfm=bw;
random intercept dNegAffectC /sub=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

Selected Output

Fixed Effects

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.0162	0.08147	999	61.57	<.0001
dNegAffectC	1.3203	0.02899	29E3	45.54	<.0001

Random Effects

Estimated G Matrix				
Row	Effect	id	Col1	Col2
1	Intercept	1	6.5180	-0.1717
2	dNegAffectC	1	-0.1717	0.3279

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	6.5180	0.2968	21.96	<.0001
UN(2,1)	id	-0.1717	0.07478	-2.30	0.0217
UN(2,2)	id	0.3279	0.03641	9.01	<.0001
SP(POW)	id	0.9685	0.002030	477.16	<.0001
Residual		3.4027	0.02883	118.01	<.0001

Estimating Between- and Within-Person Effects in the Same Model

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{dNA}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + \gamma_{01}(\mathbf{mNA}_{j}) + u_{0j}$

$\beta_{1j} = \gamma_{10} + u_{1j}$

The dNA slope asks whether the same person experiences an increase in craving when NA is high

The mNA effect asks whether people with higher average NA also have higher average craving

Mixed model for Between and Within NA on Craving

$$\textbf{CRAV}_{ij} = \gamma_{00} + \gamma_{01}(\textbf{mNA}_j) + \gamma_{10}(\textbf{dNA}_{ij}) + u_{0j} + u_{1j}(\textbf{dNA}_{ij}) + e_{ij}$$

γ_{00} : fixed intercept
γ_{01} : between-person NA effect on intercept
γ_{10} : within-person NA slope
u_{0j} : random intercept
u_{1j} : random within-person NA slope
e_{ij} : residual

SAS Syntax

```
title "between & within association between NA and craving, with random slope";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= mNegAffectC dNegAffectC /s ddfm=bw;
random intercept dNegAffectC /subject=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

Selected Output

Fixed Effects

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.0116	0.07808	998	64.19	<.0001
mNegAffectC	1.9503	0.2097	998	9.30	<.0001
dNegAffectC	1.3149	0.02909	29E3	45.20	<.0001

Random Effects

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	5.9770	0.2726	21.92	<.0001
UN(2,1)	id	-0.1135	0.07122	-1.59	0.1109
UN(2,2)	id	0.3325	0.03649	9.11	<.0001
SP(POW)	id	0.9685	0.002034	476.25	<.0001
Residual		3.4020	0.02882	118.05	<.0001

PennState

Interpreting the fixed effects output

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.0116	0.07808	998	64.19	<.0001
mNegAffectC	1.9503	0.2097	998	9.30	<.0001
dNegAffectC	1.3149	0.02909	29E3	45.20	<.0001

Between (mNegAffect): People with higher mean NA experience higher mean craving in daily life compared to people with lower mean NA.

Within (dNegAffect): Compared to themselves, people experience increased craving during high NA versus low NA moments.

Remember, relationship is contemporaneous

- We can get the same story using craving as a predictor of negative affect
- Reminds us that these associations are correlational
- And don't establish which predicts which

Selected Output from Craving as a Predictor of NA

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1.5278	0.01135	998	134.65	<.0001
mCravingC	0.03824	0.003854	998	9.92	<.0001
dCravingC	0.08676	0.002503	29E3	34.66	<.0001

- Same conclusions:
  - Higher mean craving is associated with higher mean NA (between-person)
  - Moments of increased craving are associated with increased NA (within-person)

Directionality via Temporal Lags

- To find out if NA is a predictor of craving, we can use NA from the previous observation to predict craving at the current observation
- We “lag” the variable in the dataset to achieve this

Syntax for Creating lags

```
proc sort data=ilddataset; by id date hour minute; run;

data ilddataset; set ilddataset;
by id date;
 dNegAffectC_lag1=lag(dNegAffectC); /*Lag NA*/
 CravingC_lag1=lag(CravingC); /*Lag Craving (Control Var)*/
if first.id | first.date then do; /*Sets overnight lags and lags across people to
missing*/
 dNegAffectC_lag1=.;
 CravingC_lag1=.;
label dNegAffectC_lag1="dNegAffect, lagged 1 obs"
 CravingC_lag1="Craving, lagged 1 obs";
run;
```

Obs	id	date	HourWithinDay	MinuteWithinHour	dNegAffectC	dNegAffectC_lag1
1	1	20164	10	1	.	.
2	1	20164	11	25	-0.53571	.
3	1	20164	15	4	0.46429	-0.53571
4	1	20164	16	58	0.46429	0.46429
5	1	20164	20	25	0.46429	0.46429
6	1	20165	9	1	0.46429	.
7	1	20165	11	25	0.46429	0.46429
8	1	20165	14	4	0.46429	0.46429
9	1	20165	16	58	0.46429	0.46429
10	1	20165	19	25	.	0.46429
11	1	20166	9	1	-0.53571	.
12	1	20166	12	25	-0.53571	-0.53571
13	1	20166	15	4	0.46429	-0.53571
14	1	20166	17	58	.	0.46429
15	1	20166	19	25	0.46429	.
16	1	20167	9	1	0.46429	.
17	1	20167	11	25	0.46429	0.46429
18	1	20167	15	4	.	0.46429
19	1	20167	16	58	0.46429	.
20	1	20167	20	25	0.46429	0.46429
21	1	20168	9	1	-0.53571	.
22	1	20168	11	25	-0.53571	-0.53571
23	1	20168	14	4	-0.53571	-0.53571
24	1	20168	17	58	.	-0.53571
25	1	20168	20	25	.	.

PennState

SAS Syntax for Lagged Model

```
title "lagged model, lagged NA predicting current craving";
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= CravingC_lag1 dNegAffectC_lag1 /s ddfm=bw;
random intercept dNegAffectC_lag1 /subject=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

We control for lagged craving (grand mean centered) to remove craving influence on itself

We test the within-person effect of lagged NA

We could include person-mean NA for between effect … but we don’t here to keep model simple

Selected Output

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	4.6859	0.07739	999	60.55	<.0001
CravingC_lag1	0.07451	0.006982	21E3	10.67	<.0001
dNegAffectC_lag1	0.2303	0.02568	21E3	8.97	<.0001

Significant lagged effect, NA predicts craving at momentary level. Lagged effect does not differ significantly between people (random slope is not significant)

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	5.8168	0.2848	20.43	<.0001
UN(2,1)	id	0.08533	0.05859	1.46	0.1453
UN(2,2)	id	0.01760	0.01921	0.92	0.1798
SP(POW)	id	0.7421	0.5324	1.39	0.1633
Residual		3.5612	0.03578	99.52	<.0001

PennState

Within-Person Process by Between-Person Characteristics

Testing Between-Person Differences in Within-Person Effects

- We know the within-person association between NA and craving is stronger for some versus others

- Can we predict who these people are?

- We hypothesize that the coupling between NA and craving may differ by level of nicotine dependence

PennState

MLM testing Dependence Effects on NA-Craving Slope

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{dNA}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_{j}) + u_{0j}$

$\beta_{1j} = \gamma_{10} + \gamma_{11}(\mathbf{DEP}_{j}) + u_{1j}$

PennState

Mixed Model: Dependence Effects on NA-Craving Slope

$$\begin{aligned}\mathbf{CRAV}_{ij} = {} & \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j) \\ & + \gamma_{10}(\mathbf{dNA}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \times \mathbf{dNA}_{ij}) \\ & + u_{0j} + u_{1j}(\mathbf{dNA}_{ij}) + e_{ij}\end{aligned}$$

γ_{00} : fixed intercept

γ_{01} : Dependence effect on intercept

γ_{10} : within-person NA slope

γ_{11} : Dependence effect on NA slope (cross-level interaction)

u_{0j} : random intercept

u_{1j} : random within-person NA slope

e_{ij} : residual

PennState

SAS Syntax

```
title "within association between NA and craving, by dependence";
proc mixed data=ILDDataset method=ml covtest;
 class id;
 model Craving= ftnd0
                         dNegAffectC dNegAffectC*ftnd0 /s ddfm=bw;
 random intercept dNegAffectC /sub=id type=un g gcorr;
 repeated /sub=id type=sp(pow)(studymins);
 estimate "NA effect, low dep" dNegAffectC 1 dNegAffectC*ftnd0 0;
 estimate "NA effect, high dep" dNegAffectC 1 dNegAffectC*ftnd0 1;
run; title;
```

Estimating Simple Slopes

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j)$$
$$+ \boxed{\gamma_{10}(\mathbf{dNA}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \text{ x } \mathbf{dNA}_{ij})}$$

Simple Slope:

$$\gamma_{10}(\mathbf{dNA_{ij}}) + \gamma_{11}(\mathbf{DEP}_j \mathbf{\ x\ dNA_{ij}})$$

$$= \gamma_{10} + \gamma_{11}(\mathbf{DEP}_j)$$

PennState

Generating simple slopes: $\gamma_{10} + \gamma_{11}(DEP_j)$

- What is the dNA effect for Low Dependence?

  Simple Slope: $\gamma_{10} + \gamma_{11}(\mathbf{0})$

```
estimate "NA effect, low dep" dNegAffectC 1 dNegAffectC*ftnd0 0;
```

- What is the dNA effect for High Dependence?

  Simple Slope: $\gamma_{10} + \gamma_{11}(\mathbf{1})$

```
estimate "NA effect, high dep" dNegAffectC 1 dNegAffectC*ftnd0 1;
```

OUTPUT:
Fixed Effects and Simple Slopes

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	4.5917	0.07483	998	61.36	<.0001
dNegAffectC	1.3722	0.03020	29E3	45.43	<.0001
ftnd0	4.2028	0.2355	998	17.85	<.0001
dNegAffectC*ftnd0	-0.5316	0.08960	29E3	-5.93	<.0001

Estimates								
Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
NA association, low dep	1.3722	0.03020	29E3	45.43	<.0001	0.05	1.3130	1.4314
NA association, high dep	0.8406	0.08436	29E3	9.96	<.0001	0.05	0.6752	1.0059



Craving by Negative Affect, BY Dependence Level
10
8
6
4
2
0
Craving
LowNA
HiNA
Negative Affect (person-mean centered)
LowDep
HiDep
b_{HiDep} =0.84
b_{LowDep} =1.37

RAISE DATA EXAMPLES

Personality and the Momentary Stress Process

- We are interested in understanding the momentary stress process for adolescents, and the role of personality in shaping this process

- Transactional models (i.e., Bolger & Zuckerman, 1995) hypothesize that personality affects this process in 2 ways:

  - By "selecting" the environments/experiences we have
  - By influencing our reactions to these environments/experiences

PennState

Neuroticism and the Adolescent Stress Process

- We decide to test the momentary process linking daily stressors (or hassles) and negative affect, and how this process differs based on adolescents' neuroticism

- We hypothesize that adolescents high in neuroticism will

1. Report a higher likelihood of experiencing hassles on any given moment (*greater exposure*)
2. Experience greater increases in negative affect when hassles are experienced (*greater reactivity*)

Conceptual Model



Variables

Neuroticism

-Person-level (level 2), measured once at baseline

-Interviewer reports (two interviewers, averaged)

-Does the adolescent seem…

1) Anxious, easily upset?

2) Calm, emotionally stable? (reverse coded)

PennState

Variables

Hassles

-Moment-level (level 1), measured twice a day

-Adolescent report via EMA

-Asked to report whether a number of stressful events occurred "since the morning" (if afternoon), or "since the afternoon" (if evening)

-Dichotomized into 1=one or more hassles occurred, 0=no hassles

Variables

Negative Affect

-Moment-level (level 1), measured three times a day

-Adolescent report via EMA

-Asked to rate on a sliding scale (0-100) how they felt "right now"across 7 negative emotion adjectives

(e.g., mad, nervous, sad, lonely, worried)

-Mean was taken at each observation

Descriptive Statistics

The MEANS Procedure

Variable	Label	N	Mean	Std Dev	Minimum	Maximum
neuroAB	mean of neuroticism across raters	9780	1.8614264	0.5754736	1.0000000	4.5000000
anyhasb	any hassles occurred	6636	0.2814949	0.4497622	0	1.0000000
negaff	momentary negative affect	9798	16.9213265	16.3341162	0	99.0000000

Conceptual Model

Are adolescents with high neuroticism more likely to experience hassles on any given day?



Modeling binary outcomes

- In our example, hassles experienced at any moment is a binary outcome (did versus did not experience)
- We use logistic multilevel modeling to test this association
- In SAS, we trade in PROC MIXED for PROC GLIMMIX, which does logistic regression in a multilevel framework

MLM for Neuroticism Effect on Hassles

LEVEL 1: Log Odds(HAS_{ij}) = β_{0j}

LEVEL 2: $\beta_{0j} = \gamma_{00} + \gamma_{01}(NEUR_j) + u_{0j}$

Neuroticism is mean centered

γ_{00} is the intercept, the predicted log odds of hassles for adolescents with average levels of Neuroticism (=0)

γ_{01} is the Neuroticism association, the increase in log odds of hassles on any given day for each 1-unit increase in Neuroticism

PennState

Mixed Model for Neuroticism Effect

Multilevel Equation:

LEVEL 1: $\text{LogOdds(HAS)}_{ij} = \beta_{0j}$

LEVEL 2: $\boxed{\beta_{0j} = \gamma_{00} + \gamma_{01}(\text{NEUR}_j) + u_{0j}}$

Mixed Equation:

$\text{LogOdds(CRAV)}_{ij} = \boxed{\gamma_{00} + \gamma_{01}(\text{NEUR}_j) + u_{0j}}$

γ_{00} : fixed intercept
γ_{01} : Dependence effect on intercept
u_{0j} : random intercept

$\text{LogOdds}(\text{CRAV}_{ij}) = \gamma_{00} + \gamma_{01}(\text{NEUR}_j) + u_{0j}$

- Note that there is no level 1 residual term
- This is because with a binary outcome, if we know the mean [p], we know the variance [p*(1-p)]
- For complicated reasons, the model-estimated variance of a binary variable *can be* larger or smaller than p*(1-p)
  - This is called over- and underdispersion, respectively
- We can model this dispersion, along with residual autocorrelation, in PROC GLIMMIX

PennState

PROC GLIMMIX Syntax

```
proc glimmix data=emahome method=mspl noitprint noclprint;
class raiseid studyminsc;
model anyhasb= neuroAB_C /link=logit dist=binomial s cl ddfm=bw;
random intercept /sub=raiseid type=un g gcorr;
random studyminsc /sub=raiseid type=sp(pow)(studyminsc) residual;
covtest /wald cl; *gives Z-tests for random effects;
nloptions tech=nrridg; *optimization technique that helps convergence;
estimate "odds for mean neuro" intercept 1 /exp cl;
estimate "OR for neuroticism --> hassles" neuroAB_C 1 /exp cl;
run;
```

Selected Output

Solutions for Fixed Effects								
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Intercept	-1.2406	0.09921	285	-12.50	<.0001	0.05	-1.4359	-1.0453
neuroAB_C	0.1950	0.1691	285	1.15	0.2496	0.05	-0.1377	0.5278

Covariance Parameter Estimates							
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	Wald 95% Confidence Bounds	
UN(1,1)	raiseid	2.4092	0.2526	9.54	<.0001	1.9817	2.9925
SP(POW)	raiseid	0.7551	0.03781	19.97	<.0001	0.6810	0.8292
Residual		0.8312	0.01494	55.63	<.0001	0.8026	0.8612

Estimates											
Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper	Exponentiated Estimate	Exponentiated Lower	Exponentiated Upper
odds for mean neuro	-1.2406	0.09921	285	-12.50	<.0001	0.05	-1.4359	-1.0453	0.2892	0.2379	0.3516
OR for neuroticism --> hassles	0.1950	0.1691	285	1.15	0.2496	0.05	-0.1377	0.5278	1.2154	0.8713	1.6953

For adolescents with average neuroticism, the odds of experiencing a hassle on any given day are 29% (OR: 0.29; 95% CI: 0.24, 0.35).

With each unit increase in neuroticism, the odds of experiencing a hassle on any given day increase by 22% (OR: 1.22, CI: 0.87, 1.70)

Conceptual Model

What is the momentary association between hassles and negative affect?



An Alternate Approach to Isolating Within-Person Associations

- Our predictor, hassle occurrence, is binary

- We could center around the person mean, but this would create funny scaling
  - For example, if an adolescent experiences hassles 25% of the time, their predictor values would be
    - 1-.25 = **.75**
    - 0-.21 = **-.25**

- **Alternatively, we could simply include person mean hassles *as a covariate***
  - This will statistically remove between-person variation, allowing estimation of the within-person effect
  - Additionally, this approach keeps the predictor in the original scale

Person-Mean Hassles

- Because hassles is a binary variable, taking the mean will give a proportion for each person (ranging from 0 to 1)

- Its associated effect will therefore be comparing adolescents who *never* experienced hassles (=0) to those who *always* experienced hassles (=1)

- To get around this, I rescale by multiplying this proportion by 100 – turning it into a percentage

- The effect is now what happens to mean negative affect with each percent point increase in hassle frequency

Hassles predicting Negative Affect

LEVEL 1: $\mathbf{NA}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{HAS}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + \gamma_{01}(\mathbf{mHAS}_{j}) + u_{0j}$

$\beta_{1j} = \gamma_{10} + u_{1j}$

$mHAS_j$ is each adolescent's mean level of hassle exposure, captured as the percent of moments hassles were experienced. It is mean centered.

HAS_{ij} is the raw, time-varying hassles indicator (0, 1). With mean hassles in the model, the effect of HAS_{ij} is a within-person effect.

Mixed model for Hassles and Negative Affect

$$\mathbf{NA}_{ij} = \gamma_{00} + \gamma_{01}(\mathbf{mHAS}_j) + \gamma_{10}(\mathbf{HAS}_{ij}) + u_{0j} + u_{1j}(\mathbf{HAS}_{ij}) + e_{ij}$$

γ_{00} : fixed intercept

γ_{01} : between-person hassles effect on intercept, controlling for today's hassles

γ_{10} : within-person hassles slope

u_{0j} : random intercept

u_{1j} : random within-person hassles slope

e_{ij} : residual

SAS syntax

```
title "momentary association between hassles and NA";
proc mixed data=emahome method=ml covtest noclprint noitprint;
class raiseid studyminsc;
model negaff=manyhasbpc anyhasb /s cl ddfm=bw;
random intercept anyhasb /subject=raiseid type=un g gcorr;
repeated studyminsc /sub=raiseid type=sp(pow)(studyminsc);
run; title;
```

Selected Output

Solution for Fixed Effects								
Effect	**Estimate**	**Standard Error**	**DF**	**t Value**	**Pr > |t|**	**Alpha**	**Lower**	**Upper**
Intercept	16.1062	0.6212	288	25.93	<.0001	0.05	14.8836	17.3289
manyhasbpc	0.2019	0.02361	288	8.55	<.0001	0.05	0.1555	0.2484
anyhasb	4.5162	0.5844	6330	7.73	<.0001	0.05	3.3707	5.6618

Covariance Parameter Estimates					
Cov Parm	**Subject**	**Estimate**	**Standard Error**	**Z Value**	**Pr Z**
UN(1,1)	raiseid	100.05	9.3533	10.70	<.0001
UN(2,1)	raiseid	-7.2275	5.8525	-1.23	0.2169
UN(2,2)	raiseid	43.1640	7.7863	5.54	<.0001
SP(POW)	raiseid	0.7877	0.03096	25.45	<.0001
Residual		123.00	2.2475	54.72	<.0001

Conceptual Model

Does the momentary association between hassles and negative affect differ by adolescents' trait neuroticism?



Neuroticism and Hassles Multilevel Model

LEVEL 1: $\mathbf{NA}_{ij} = \beta_{0j} + \beta_{1j}(\mathrm{HAS}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + \gamma_{01}(\mathrm{NEUR}_j) + \gamma_{02}(\mathrm{mHAS}_j) + u_{0j}$

$\beta_{1j} = \gamma_{10} + \gamma_{11}(\mathrm{NEUR}_j) + u_{1j}$

Neuroticism and Hassles Mixed Model

$$\mathbf{NA}_{ij} = \gamma_{00} + \gamma_{01}(\text{NEUR}_j)\ \gamma_{02}(\text{mHAS}_j)$$
$$+ \gamma_{10}(\text{HAS}_{ij}) + \gamma_{11}(\text{NEUR}_j \text{ x } \text{HAS}_{ij})$$
$$+ u_{0j} + u_{1j}(\text{HAS}_{ij}) + e_{ij}$$

γ_{00} : fixed intercept

γ_{01}: between-person effect of neuroticism on intercept

γ_{02} : between-person hassles effect on intercept, controlling for today's hassles

γ_{10} : within-person hassles slope

γ_{11}: between-person effect of neuroticism on hassles slope

u_{0j} : random intercept

u_{1j} : random within-person hassles slope

e_{ij} : residual

Model for Neuro x Hassles

```
title "Neuroticism x hassles predicting NA";
proc mixed data=emahome method=ml covtest noclprint noitprint;
class raiseid studyminsc;
model negaff=manyhasbpc neuroAB_C anyhasb neuroAB_C*anyhasb /s cl
ddfm=bw;
random intercept anyhasb /subject=raiseid type=un g gcorr;
repeated studyminsc /sub=raiseid type=sp(pow)(studyminsc);
*simple slopes;
estimate "hassles-->NA slope, low Neuroticism" anyhasb 1 neuroAB_C*anyhasb -0.6;
estimate "hassles-->NA slope, high Neuroticism" anyhasb 1 neuroAB_C*anyhasb 0.6;
run; title;
```

Estimating simple slopes

$$\mathbf{NA}_{ij} = \gamma_{00} + \gamma_{01}(\mathrm{NEUR}_j)\ \gamma_{02}(\mathrm{mHAS}_j)$$
$$+ \boxed{\gamma_{10}(\mathrm{HAS}_{ij}) + \gamma_{11}(\mathrm{NEUR}_j \text{ x } \mathrm{HAS}_{ij})}$$

Simple Slope:

$$\gamma_{10}(\mathrm{HAS}_{ij}) + \gamma_{11}(\mathrm{NEUR}_j \text{ x } \mathrm{HAS}_{ij})$$

$$= \gamma_{10} + \gamma_{11}(\mathrm{NEUR}_j)$$

Generating simple slopes: $\gamma_{10} + \gamma_{11}(NEUR_j)$

- What is the within hassles effect for Low Neuroticism?
  - Hold Neur at 1 SD below the Mean (=-0.6)

Simple Slope: $\gamma_{10} + \gamma_{11}(\mathbf{-0.6})$

estimate “Hassles Slope, Low Neur" anyhasb **1** neuroAB_C*anyhasb -**0.6**;

- What is the within hassles effect for High Neuroticism?
  - Hold Neur at 1 SD below the Mean (=0.6)

Simple Slope: $\gamma_{10} + \gamma_{11}(\mathbf{0.6})$

estimate “Hassles Slope, Low Neur" anyhasb **1** neuroAB_C*anyhasb **0.6**;

PennState

OUTPUT:
Fixed Effects and Simple Slopes

Solution for Fixed Effects								
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Intercept	16.1903	0.6253	284	25.89	<.0001	0.05	14.9596	17.4210
manyhasbpc	0.2014	0.02372	284	8.49	<.0001	0.05	0.1547	0.2481
neuroAB_C	0.1867	1.0515	284	0.18	0.8592	0.05	-1.8829	2.2564
anyhasb	4.4024	0.5846	6277	7.53	<.0001	0.05	3.2563	5.5484
neuroAB_C*anyhasb	-0.4269	1.0254	6277	-0.42	0.6772	0.05	-2.4369	1.5832

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > |t|
hassles-->NA slope, low Neuroticism	4.6585	0.8546	6277	5.45	<.0001
hassles-->NA slope, high Neuroticism	4.1463	0.8427	6277	4.92	<.0001

PennState


Negative Affect by Hassles, BY Neuroticism
Negative Affect
30
28
26
24
22
20
18
16
14
12
10
NoHassles
Hassles
Any Hassles Occurred
LowNeur
HiNeur

EXTRA SLIDES

Brief Intro to SAS Programming

Brief introduction to SAS's setup

- Three important windows
  - Editor
    - Window where you write your programs
  - Log
    - Window where you are informed of what SAS is doing when it runs your program
    - Window where you check for errors in your program
  - Output
    - Window where you see the output (results) from your program

Program management and organization

- For the “editor,” “log,” and “output” windows
  - Save as you normally would in a Windows-based program
    - File - Save As…
  - Print as you normally would in a Windows-based program
    - File - Print
  - May also “copy” and “paste” from these windows into Word documents
- File extension for SAS programs is “.sas”

Data management and organization

- SAS uses “libraries” to organize and save data
  - Default library is “work”
    - Does not save datasets permanently, only a “working” directory with “working” datasets in the current SAS session
    - When you close SAS, datasets in “work” are lost
  - You may make a library that points to a location on your computer (or external drive, etc.) where you have datasets saved (or want to have datasets saved)
    - Datasets may be “read from” and “written to” that library, which will open the dataset from, or save the dataset to, the specified location on your computer
- File extension for SAS datasets is “.sas7bdat”

Data management and organization

- May view actual dataset within SAS
  - In “explorer” window:
    - Double-click “libraries”
    - Double-click the library you want to view
    - Double-click the dataset you want to view
- Missing data has a special code
  - “.”

Writing and running a program

- Comments
  - *write comment here;
  - /*write comment here*/
- Run
  - Highlight, click on the little “running man” icon on the tool bar located across the top of the SAS window
  - Or, Highlight and press F3

Useful “procedures” for data exploration

- PROC CONTENTS
  - Produces a list of all variables in specified dataset

```
PROC CONTENTS DATA = EXAMPLE;
RUN;
```

Useful "procedures" for data exploration

- PROC FREQ
  - Produces frequency tables for specified variables

```
PROC FREQ DATA = EXAMPLE;
  TABLES GENDER;
RUN;
```

Useful "procedures" for data exploration

- PROC UNIVARIATE
  - Produces a variety of descriptive statistics for specified variables
  - NORMAL option produces normal probability plots
  - PLOT option produces stem-and-leaf plots and boxplots

```
PROC UNIVARIATE DATA = EXAMPLE PLOT;
VAR GGPA VGRE;
RUN;
```

Useful "procedures" for data exploration

- PROC MEANS
  - Produces smaller list of descriptive statistics for specified variables

```
PROC MEANS DATA = EXAMPLE;
V VAR IQ CGPA;
RUN;
```

Example of 3-way interaction

A "Big" Predictive Model

- NA seems to be more strongly linked with craving for those with Low versus High Dependence

- But perhaps this pattern differs based on background factors

- Consider childhood adversity. Does dependence matter as much for those with more adverse backgrounds?

Childhood Adversity

- A count of adverse experiences (parental divorce, domestic violence, poverty) experienced in childhood



Adversity x Dependence x NA

- We hypothesize a three-way interaction between Childhood Adversity, Dependence, and Momentary NA in predicting smoking
- We think that Dependence will strengthen the NA-Craving Coupling for those with Low Adversity
- But not so much for those with High Adversity; dependence may matter less

MLM for 3-way interaction

LEVEL 1: $\mathbf{CRAV}_{ij} = \beta_{0j} + \beta_{1j}(\mathbf{dNA}_{ij}) + e_{ij}$

LEVEL 2: $\beta_{0j} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j) + \gamma_{02}(\mathbf{ADV}_j)$
$+ \gamma_{03}(\mathbf{DEP}_j \text{ x } \mathbf{ADV}_j) + u_{0j}$

$\beta_{1j} = \gamma_{10} + \gamma_{11}(\mathbf{DEP}_j) + \gamma_{12}(\mathbf{ADV}_j)$
$+ \gamma_{13}(\mathbf{DEP}_j \text{ x } \mathbf{ADV}_j) + u_{1j}$

Mixed Equation for 3-way interaction

Intercept

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j) + \gamma_{02}(\mathbf{ADV}_j) + \gamma_{03}(\mathbf{DEP}_j \times \mathbf{ADV}_j)$$

dNA Slope

$$+ \gamma_{10}(\mathbf{dNA}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \times \mathbf{dNA}_{\mathbf{ij}}) + \gamma_{12}(\mathbf{ADV}_j \times \mathbf{dNA}_{\mathbf{ij}})$$

$$+ \gamma_{13}(\mathbf{DEP}_j \times \mathbf{ADV}_j \times \mathbf{dNA}_{\mathbf{ij}})$$

Random Effects

$$+ u_{0j} + u_{1j}(\mathbf{dNA}_{ij}) + e_{ij}$$

SAS Syntax

```
proc mixed data=ILDDataset method=ml covtest;
class id;
model Craving= ftnd0 ChAdv ftnd0*ChAdv
              dNegAffectC dNegAffectC*ftnd0 dNegAffectC*ChAdv
              dNegAffectC*ChAdv*ftnd0 /s ddfm=bw;
random intercept dNegAffectC /sub=id type=un g gcorr;
repeated /sub=id type=sp(pow)(studymins);
run; title;
```

Selected Output

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	3.3828	0.08301	996	40.75	<.0001
ftnd0	4.0966	0.5028	996	8.15	<.0001
ChAdv	0.8914	0.04109	996	21.69	<.0001
ftnd0*ChAdv	-0.6001	0.1100	996	-5.46	<.0001
dNegAffectC	1.2536	0.04001	29E3	31.33	<.0001
ftnd0*dNegAffectC	0.3983	0.2351	29E3	1.69	0.0902
ChAdv*dNegAffectC	0.08961	0.01942	29E3	4.62	<.0001
ftnd0*ChAdv*dNegAffe	-0.2697	0.05164	29E3	-5.22	<.0001

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	id	3.2858	0.1523	21.58	<.0001
UN(2,1)	id	-0.1169	0.05155	-2.27	0.0233
UN(2,2)	id	0.2763	0.03346	8.26	<.0001
SP(POW)	id	0.9682	0.002093	462.49	<.0001
Residual		3.4013	0.02880	118.10	<.0001

Unpacking the 3-Way Interaction

- We want to know what the Dependence x NA interaction looks like for low versus high adversity

- We pick two values in the adversity scale
  - Low Adversity: 0 Adverse events
  - High Adversity: 4 adverse events

- And generate model predictions based on these

Select terms for DEP x dNA

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j) + \gamma_{02}(\mathbf{ADV}_j) + \gamma_{03}(\mathbf{DEP}_j \times \mathbf{ADV}_j)$$
$$+ \gamma_{10}(\mathbf{dNA}_{ij}) + \boxed{\gamma_{11}(\mathbf{DEP}_j \times \mathbf{dNA}_{\mathbf{ij}})} + \gamma_{12}(\mathbf{ADV}_j \times \mathbf{dNA}_{\mathbf{ij}})$$
$$+ \boxed{\gamma_{13}(\mathbf{DEP}_j \times \mathbf{ADV}_j \times \mathbf{dNA}_{\mathbf{ij}})}$$

Simple Interaction:

$$\gamma_{11}(\mathbf{DEP}_j \times \mathbf{dNA}_{\mathbf{ij}}) + \gamma_{13}(\mathbf{DEP}_j \times \mathbf{ADV}_j \times \mathbf{dNA}_{\mathbf{ij}})$$

$$= \gamma_{11} + \gamma_{13}(\mathbf{ADV}_j)$$

Simple Interaction: $\gamma_{11} + \gamma_{13}(ADV_j)$

- What is the Dep x NA interaction for Low Adversity (=0 Adverse Events)?

estimate “Dep x NA, Low Adversity“
dNegAffectC*ftnd0 **1** dNegAffectC*ChAdv*ftnd0 **0**;

- And for High Adversity (=4 Adverse Events)?

estimate “Dep x NA, High Adversity"
dNegAffectC*ftnd0 **1** dNegAffectC*ChAdv*ftnd0 **4**;

Output

Estimates						
Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha
Neg Aff x Dependence, Low Adversity	0.3983	0.2351	29E3	1.69	0.0902	0.05
Neg Aff x Dependence, High Adversity (4)	-0.6806	0.1045	29E3	-6.52	<.0001	0.05

Contrary to our hypothesis, Dependence seems to be a significant moderator of the NA-craving coupling for High but not Low Adversity

Let's unpack this further to see what's going on…

Simple dNA slopes

$$\mathbf{CRAV}_{ij} = \gamma_{00} + \gamma_{01}(\mathbf{DEP}_j) + \gamma_{02}(\mathbf{ADV}_j) + \gamma_{03}(\mathbf{DEP}_j \text{ x } \mathbf{ADV}_j)$$
$$\boxed{+ \gamma_{10}(\mathbf{dNA}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \text{ x } \mathbf{dNA_{ij}}) + \gamma_{12}(\mathbf{ADV}_j \mathbf{\ x\ dNA_{ij}}) + \gamma_{13}(\mathbf{DEP}_j \text{ x } \mathbf{ADV}_j \mathbf{\ x\ dNA_{ij}})}$$

Simple Slope:

$$\gamma_{10}(\mathbf{dNA}_{ij}) + \gamma_{11}(\mathbf{DEP}_j \text{ x } \mathbf{dNA_{ij}}) + \gamma_{12}(\mathbf{ADV}_j \mathbf{\ x\ dNA_{ij}})$$
$$+ \gamma_{13}(\mathbf{DEP}_j \text{ x } \mathbf{ADV}_j \mathbf{\ x\ dNA_{ij}})$$

$$= \gamma_{10} + \gamma_{11}(\mathbf{DEP}_j) + \gamma_{12}(\mathbf{ADV}_j) + \gamma_{13}(\mathbf{DEP}_j \text{ x } \mathbf{ADV}_j)$$

Simple Slopes:

$$\gamma_{10} + \gamma_{11}(DEP_j) + \gamma_{12}(ADV_j) + \gamma_{13}(DEP_j \text{ x } ADV_j)$$

```
estimate "Neg Aff, Low Dependence Low Adversity"
         dNegAffectC 1 dNegAffectC*ftnd0 0
         dNegAffectC*ChAdv 0 dNegAffectC*ChAdv*ftnd0 0;

estimate "Neg Aff, Low Dependence High Adversity"
         dNegAffectC 1 dNegAffectC*ftnd0 0
         dNegAffectC*ChAdv 4 dNegAffectC*ChAdv*ftnd0 0;

estimate "Neg Aff, High Dependence Low Adversity"
         dNegAffectC 1 dNegAffectC*ftnd0 1
         dNegAffectC*ChAdv 0 dNegAffectC*ChAdv*ftnd0 0;

estimate "Neg Aff, High Dependence High Adversity"
         dNegAffectC 1 dNegAffectC*ftnd0 1
         dNegAffectC*ChAdv 4 dNegAffectC*ChAdv*ftnd0 4;
```

Simple Slopes Output

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Neg Aff, Low Dependence Low Adversity	1.2536	0.04001	29E3	31.33	<.0001
Neg Aff, Low Dependence High Adversity	1.6121	0.05879	29E3	27.42	<.0001
Neg Aff, High Dependence Low Adversity	1.6519	0.2316	29E3	7.13	<.0001
Neg Aff, High Dependence High Adversity	0.9314	0.08634	29E3	10.79	<.0001

Adversity appears to heighten the link for those with low dependence, but dampen the link for those with high dependence


Craving = Negative Affect X Dependence Level, BY Childhood Adversity
HiAdv
LowAdv
Craving
10
8
6
4
2
0
LowNA
HiNA
LowNA
HiNA
Negative Affect (person-mean centered)
LowDep
HiDep
For High Adversity:
Low Dep has steeper slope, but High Dep has higher craving at both low and high NA
For Low Adversity:
High Dep has higher craving at low and high NA *and* steeper slope

PennState











HELPFUL REFERENCES!